COMUNICACIONES DEL CIMAT

Presentado en el 4º Coloquio de Matemáticas, Taxco, Gro., agosto de 1985.

CENTRO DE INVESTIGACION EN MATEMATICAS

Apartado Postal 402
Guanajuato, Gto.
México
Tels. (473) 2-25-50
2-02-58

Uso de la computadora en la enseñanza de las matemáticas

Alfinio Flores Peñafiel
Centro de Investigación en Matemáticas

4 º Coloquio de Matemáticas
Taxco, Gro. 21 de agosto de 1985

Introducción

El uso de la computadora puede modificar profundamente la práctica de la enseñanza de las matemáticas.
La computadora puede ser un medio para:

- lograr una enseñanza de las matemáticas más balanceada
- establecer una conexión entre la teoría y la práctica
- lograr la unificación de las matemáticas puras y aplicadas.

Sin embargo, muchas veces el cambio no es en beneficio de los alumnos, ni de las matemáticas; sino que por el contrario, sucede que el medio se convierte en un fin en sí mismo y el aprendizaje de la computación desplaza al de las matemáticas. Frecuentemente los cursos de computación son impartidos por los maestros de matemáticas durante el tiempo asignado a dicha materia.

A pesar del enorme potencial que ofrece esta herramienta, no es claro cómo y de qué forma aprovecharla. No hay consenso acerca de cuál es la manera de utilizar la computadora para lograr una mejor comprensión de los conceptos y el dominio de las técnicas y habilidades matemáticas. Hay una cantidad de software comercial, con una gran variedad de imágenes y sonidos pero con una gran pobreza conceptual. También hemos observado qué poco se ha podido integrar el uso de la computadora en la enseñanza de las diferentes materias. La enseñanza de la computación se hace sin conexión con las otras asignaturas. Sin embargo, existen ejemplos aislados donde el uso de la computadora se hace de manera integrada (ver Smith, 1984). En algunos casos existen complementos para el libro de texto (Barker y Ward, 1984).

Si bien es cierto que la presencia de la computadora es cada día más frecuente en las escuelas y universidades de nuestro país, la mera accesibilidad a estas máquinas no garantiza su uso adecuado.

Es necesario determinar cuáles son los conceptos fundamentales de las matemáticas, cuáles deben ser enseñados, cuáles de éstos pueden ser mejor tratados con la computadora y de qué manera. Con el acceso de practicamente todos los alumnos a la computadora en un futuro cercano, es también necesario reexaminar el énfasis que se hace en la enseñanza. Habilidades a las que se dedica mucha atención en el presente, serán sustituidas por otras. (Conference Board for the Mathematical Sciences,1982; 1984)
Por ejemplo, programar la computadora es una oportunidad magnífica para desarrollar el pensamiento algorítmico, así como para aprender a plantear y resolver problemas.

La preparación de los nuevos matemáticos y profesores de matemáticas debe también reflejar estos cambios. Hay que dar el tiempo y énfasis necesarios para que aprendan cómo usar la computadora para la enseñanza de las matemáticas.

Asímismo, la computadora es capaz de ahorrarnos una cantidad enorme de trabajo. Es evidente que en los cursos de matemáticas se debe aprovechar esta capacidad. Los paquetes de estadística, los de álgebra lineal, los de soluciones numéricas de ecuaciones diferenciales son ejemplos claros. Este es uno de los usos más importantes de la computadora, aunque no se discutirá en el presente trabajo.

A continuación daré algunos ejemplos de cómo la computadora puede ser utilizada dentro de algunos cursos de matemáticas, tratando que cumplan con las siguientes características:

- el alumno tiene un papel activo;
- el alumno le dice a la máquina qué debe hacer, la utiliza como herramienta o le enseña a hacer algo (ver Taylor, 1980);
- se tratan conceptos importantes de las matemáticas
- no necesitan equipo muy sofisticado y costoso;
- pueden realizarse en condiciones reales;

Daré tres tipos de ejemplos:
- primero: los alumnos escriben sus propios programas;
- segundo: los alumnos utilizan un programa escrito por otros para ilustrar o explorar algún concepto;
- tercero: uso de paquetes "inteligentes".

1 PROGRAMAS ESCRITOS POR LOS ALUMNOS

Algunos investigadores han encontrado paralelismos entre el tipo de pensamiento que se requiere para esctibir, probar, corregir y refinar un programa y varios aspectos del pensamiento matemático. Hatfield (1982) menciona los siguientes como algunos procesos del pensamiento matemático que se pueden beneficiar por programar una computadora:

analizar
simplificar
particularizar
generalizar
justificar
conjeturar
estructurar

Los alumnos pueden escribir programas sencillos y no muy largos que ilustren algunos conceptos claves de los cursos. El tener que explicar a la computadora con precisión cómo utilizar el concepto, es una de las mejores formas de lograr en el alumno una mejor comprensión. Pequeños programas pueden ilustrar grandes ideas. El siguiente programa grafica líneas rectas a través del origen (la computadora Sinclair coloca el origen de coordenadas en la parte inferior izquierda de la pantalla) :

```
5  INPUT " pendiente " ; A
10 FOR X = 0 TO 255
15 LET Y = A * X
20 PLOT X , Y
30 NEXT X
```

Los alumnos con este programa están trabajando con las ideas fundamentales de la geometría analítica, que son la de asociar puntos en el plano con parejas ordenadas de números, la de hacer corresponder curvas en el plano a ecuaciones de dos variables. Además, en este caso se familiarizan también con el concepto de pendiente de una recta.

Se dan ejemplos para valores de la pendiente de 0, 0.5, 1, 2, 4.

Para escribir un programa que grafique con el origen en el centro de la pantalla, o para cambiar de escala, los alumnos tendrán que entender las transformaciones tales como traslaciones y homotecias. El programa GRAFICAS 1 (ver apéndice) traza la gráfica de la función y = x + 2. Con sólo modificar un renglón (el 30) se puede estudiar una gama muy amplia de funciones. Además para cada función se puede ver el resultado de graficarla con distintas escalas.

Programas para cálculo.

El método de incrementos juega un papel fundamental para comprender el concepto de derivada. El siguiente programa calcula la pendientes de secantes a una curva, para intervalos cada vez más pequeños.

```
10 DEF FN Y(X) = X * X
20 INPUT " punto inicial " ; X
30 INPUT " incremento "; H
40 FOR N = 1 TO 8
50 LET S = ( FN Y(X+H) - FN Y(X) ) / H
60 PRINT H , S
70 LET H = H / 4
80 NEXT N
```

Un alumno que escriba este programa tendrá que prestar atención cuidadosa al cociente de las diferencias para el renglón clave del programa.

Se dan dos ejemplos de resultados de este programa, ambos para un valor de x = 3, con un incremento positivo en un caso y uno negativo en el otro.

Derivación numérica

Este programa grafica una función f(x) y la aproximación a la derivada de la función dada por el cociente de los incrementos

$$\frac{f(x+h) - f(x)}{h}$$

```
20  DEF FN Y (X) = X*X
30  INPUT "incremento: "; H
40  GOSUB 300
50  FOR X=-3.125 TO 3.25 STEP .0125
60     LET Y = FN Y(X)
70     GOSUB 200
80     LET Y = ( FN Y(X+H) - FN Y(X) ) / H
90     GOSUB 200
100 NEXT X
110 STOP
200 REM subrutina para graficar ***
300 REM *** subrutina para ejes ***
```

Las subrutinas para graficar y para los ejes son las mismas que las del programa GRAFICAS 1 (ver apéndice)

Se anexan ejemplos para las funciones $f(x) = x^2$ y para $f(x) = sen(x)$

(Para más ejemplos de programas que los alumnos pueden escribira para cálculo ver Smith,1984; Sagan,1984; Flores Peñafiel,1984)

2 LOS ALUMNOS USAN PROGRAMAS YA HECHOS

Los alumnos pueden utilizar programas ya hechos para explorar y repasar conceptos matemáticos. En esta modalidad no es necesario que los alumnos comprendan todo el programa. Sólo es necesario que entiendan la parte del programa relacionada directamente con el concepto en cuestión, a fin de que puedan hacer modificaciones sencillas al programa, y en ocasiones ni siquiera tendrán que modificar el programa, sino sólo cambiar datos o parámetros. Esto les dará oportunidad de ver muchos ejemplos, ver qué pasa con las modificaciones, tratar de encotrar generalizaciones a partir de los casos vistos. La computadora se puede convertir así en el corazón de un enfoque "experimental" de las matemáticas. El programa SISTEMAS I es un ejemplo de un programa que forma parte de un paquete de actividades para nivel medio superior, que formarán un laboratorio de matemáticas.

Simulación

Representar en forma gráfica o visual un fenómeno o proceso ha sido siempre una de las mejores maneras de lograr una mejor comprensión de este. Muchas veces aún con programas cortos se pueden simular procesos (ver por ejemplo Flores, 1984). La computadora tiene la capacidad de representar procesos dinámicos, de representar cambios conforme transcurre el tiempo. Este es uno de los aspectos que la imagen estática de un libro no puede dar.

Por ejemplo, una función de $\mathbb{R} \to \mathbb{R}^2$ puede tener la misma imagen que otra, sin embargo recorrerla con velocidades distintas, como $f(t) = (t^3, t^3)$ y $g(t) = (t,t)$ que representan la misma recta, pero ésta se recorre de manera diferente al hacer variar t. El programa ECPARAMETR grafica ecuaciones en forma paramétrica. Al ejecutarlo los alumnos pueden percibir la diferencia de velocidades al recorrer la recta de una forma o de otra. El hecho de que la computadora grafique un conjunto discreto de puntos puede así tener ventajas didácticas.

Al estudiar el movimiento planetario no sólo importa la forma de la órbita, sino cómo es recorrida por el planeta. Este no se mueve alrededor del sol con una rapidez constante, sino que se mueve más rápido cerca del sol. Esto es más notorio mientras mayor sea la excentricidad de la órbita. Un ejemplo particularmente atractivo en este año es el cometa Halley. El programa KEPLERLEY2 simula el movimiento de un cometa en su órbita de modo que se satisfaga la segunda ley de Kepler. (Adaptado del programa "Kepler" (Tattersfield, 1984)).







3 EL USO DE PAQUETES "INTELIGENTES"

Hay diversos paquetes con los cuales se pueden enfatizar diferentes conceptos matemáticos. Entre ellos tal vez los más conocidos sean las hojas electrónicas (por ejemplo VisiCalc). Arganbright (1984) da algunas ideas de cómo utilizarlas para la enseñanza de las matemáticas. Tiene también un libro de cómo utilizarlas para técnicas tales como el método de Newton, polinomios de Taylor, valores propios, la regla de Bayes, el algoritmo simplex y otras más (Arganbright, 1985).

Existen otros paquetes tales como TK!Solver que resuelven ecuaciones de modo similar a como se hace en álgebra. (Para una reseña de este paquete ver Williams, 1982). De acuerdo a este autor, paquetes como este tienen el poetencial de convertirse para el álgebra en lo que las calculadoras son para la aritmética, un medio rápido y preciso de obtener los resultados.

Las computadoras no sólo son capaces de hacer operaciones numéricas, sino que también pueden manipular expresiones simbólicas y dar respuestas con símbolos. Aunque los sistemas simbólicos han existido para computadoras grandes desde hace muchos años (por ejemplo el sistema MACSYMA), hay ahora también paquetes para micro computadoras. Comentaré brevemente acerca de uno de estos, el paquete muMATH, que entre otras cosas hace operaciones aritméticas exactas con números enteros, racionales, algebraicos y complejos, con números enteros hasta de 611 cifras. Por ejemplo puede simplificar fracciones

```
? 6/8;
@ 3/4
```

Opera con complejos:

```
? (3 + 2 * #I) * (6 - #I);
@20 + 9 * #I
```

Puede encontrar máximo común divisor entre dos números:

```
?GCD (24,60);
@ 12
```

Encontrar factorial de números y elevarlos a una potencia:

```
? 10! ^ 10;
@395940866122425193243875570782668457763038822400000000000000000000
```

Puede encontrar límites para los que muchas veces recurrimos a la regla de L'Hôpital, tal como el límite de x /sen(x) cuando $x \rightarrow 0$

```
? LIM ( X /SIN (X), X, 0);
@1
```

El sistema también resuelve integrales como $\int cx^2 + x\ \mathrm{sen}(x^2)\ dx$.

```
? INT (C * X ^ 2 + X * SIN (X ^ 2), X);
@ C * X/3 - COS (X ^2)/2
```

Las nuevas versiones de este sistema incluso resuelven ecuaciones diferenciales. (Ver Squire,1985 para una descripción de las capacidades de este sistema). En algunas universidades ya se usan estos sistemas. Francis Moon, jefe del Departamento de Mecánica Teórica y Aplicada de la Universidad de Cornell, afirma que esta nueva tecnología de software tiene la potencialidad para introducir otra vez el análisis matemático en la investigación industrial junto con el ahora popular CAD [diseño por computadora] y otros métodos numéricos de análisis (SIAM NEWS julio de 1985). Wilf(1982) plantea algunas interrogantes acerca del impacto de estos sistemas en la enseñanza de las matemáticas en la universidad. ¿Nos pasará algo semejante a lo que a muchos maestros de primaria, que todas las operaciones que enseñaban a hacer en largos y penosos meses, ahora un niño es capaz de aprender en unos minutos con una calculadora? ¿Nos opondremos en las universidades al uso de estos sistemas, así como en tantas primarias y secundarias se oponen a que los alumnos usen las calculadoras?

Referencias

Argonbright, Deane. Mathematical applications of an electronic spreadsheet." En <u>Computers in mathematics education. 1984 Yearbook</u>, National Council of Teachers of Mathematics, 1984. p. 184-193.

Argonbright, Deane. <u>Mathematical applications of electronic spreadsheets.</u> McGraw Hill, 1985.

Barker, W.H.; Ward, J.E. <u>The calculus companion to accompany Calculus 2nd ed by Howard Anton.</u> Wiley, 1984.

Conference Board of the Mathematical Sciences. <u>The mathematical sciences curriculum K-12: what is still fundamental and what is not.</u> National Science Foundation, 1982.

Conference Board of the Mathematical Sciences. <u>New goals for mathematical sciences education</u>. Conference Board of the Mathematical Sciences, 1984.

Fey, James T.(ed) <u>Computing and mathematics</u>. National Council of Teachers of Mathematics, 1984.

Flores, Alfinio. "A microcomputer and the law of small numbers." <u>Arithmetic Teacher, 31</u>(7), (marzo 1984) 60-61.

Flores Peñafiel, Alfinio. "La microcomputadora en la enseñanza del cálculo". Trabajo presentado en el Congreso Anual de la Sociedad Matemática Mexicana, Mérida, noviembre 1984.

<u>Guidelines and standards for the education of secondary school teachers of science and mathematics.</u> American Association for the Advancement of Science, 1971.

Hatfield, Larry. L. "Instructional computing in mathematics teacher education." <u>Journal of Research Development in Education, 15</u> (4),(1982) 30-44.

<u>The muMATH/mu SIMP-80 Symbolic Mathematics System Reference Manual for the Apple II Computer</u>. The Soft Warehouse, 1981.

News Briefs "Cornell will use MACSYMA and muMATH." SIAM News, 18(4), julio 1985, 2.

Recommendations on the mathematical training of teachers. Mathematical Association of America. 1983.

Recommendations for a general mathematical sciences program. Mathematical Association of America. 1981.

Sagan, Hans. Calculus accompanied on the Apple. Reston, 1984.

Smith, David A. Interface: calculus and the computer 2a ed. Saunders College Publishing, 1984.

Squire, William. "muMATH system effective tool for algebra." SIAM News, 17(6), (noviembre 1984) 4.

Tattersfield, Donald. Halley's comet. Blackwell, 1984.

Wilf, Herbert S. "The disk with the college education." American Mathematical Monthly 89 (1982), 4-8.

Taylor, R. P. The computer in the school: tutor, tool, tutee. New York, Teachers College Pr., 1980.

Williams, Gregg. "Software Arts' TK! Solver." Byte, octubre 1982, 360-76.

PROGRAMA PARA GRAFICAR CON COORDENADAS CARTESIANAS

```
5 REM GRAFICAS 1
8  REM COORDENADAS CARTESIANAS
10 GOSUB 100: REM grafica ejes
15 REM ************************************
20 REM programa principal
20 FOR X = -3.2 TO 3.175 STEP .025
30 LET Y = X*X
40 GOSUB 200
50 NEXT X
99 STOP
100 REM ******* subrutina para ejes **************
110 PLOT 128 , 0 : DRAW 0 , 175
120 PLOT 0 , 88 : DRAW 255 , 0
130 RETURN
200 REM ************************************
205 REM subrutina para cambio de coordenadas y graficar
210 LET H = 40 * X + 128
220 LET V = 40 * Y + 88
230 IF V < 0 OR V > 175 THEN RETURN
240 PLOT H , V
250 RETURN
```

PROGRAMA PARA GRAFICAR ECUACIONES PARAMETRICAS

```
5   REM ECPARAMETR
10 GOSUB 100: REM grafica ejes
15 REM ************************************
30 REM programa principal
40 FOR T = -3.2 TO 3.175 STEP .025
50 LET X = T * T * T
60 LET Y = T * T
70 GOSUB 200 : REM grafica punto
80 NEXT X
99 STOP
100 REM ******* subrutina para ejes *************
110 PLOT 128 , 0 : DRAW 0 , 175
120 PLOT 0 , 88 : DRAW 255 , 0
130 RETURN
200 REM ************************************
205 REM subrutina para cambio de coordenadas y graficar
210 LET H = 40 * X + 128
230 IF H < 0 OR H > 255 THEN RETURN
220 LET V = 40 * Y + 88
230 IF V < 0 OR V > 175 THEN RETURN
240 PLOT H , V
250 RETURN
```

2a LEY DE KEPLER

```
5  REM KEPLERLEY2
10 PRINT AT 3,0 ; "SIMULACION DE UN COMETA DE PERIODO 12 años,
   EXCENTRICIDAD 0.7, POSICIONES CADA 3 MESES"
15 LET a = 14 : REM semieje mayor
20 LET e = .7 : REM excentricidad
30 LET pp = 1986 : REM paso por prihelio
40 LET ti = 1980 : REM tiempo inicial (año)
50 LET ma = 30 : REM movimiento anual (grados por año)
60 LET u = 48 : REM número de datos
70 LET it = 0.25 : REM intervalo de tiempo (años)
80 DIM k(u+1) : DIM l(u+1) : DIM r(u+1)
90 DIM x(u+1) : DIM y(u+1)
100 LET q = a * (1- e)
110 LET b = a * SQR (1 - e*e)
120 LET rad = PI / 180
130 LET teta = 0
140 REM *********** calcula posiciones ***************
150 FOR n = 1 TO u
160     LET tetac = ma * ( ti + (n -1) * it - pp )* rad
170     LET dif = tetac - ( teta - e * SIN teta )
180     LET teta = teta + dif / ( 1- e * COS teta)
190     IF ABS (dif) > .0001 THEN GO TO 170
200        REM **** convierte polares a cartesianas *****
210        LET x(n) = a * (COS teta - e)
220        LET y(n) = b * SIN teta
230 NEXT n
240 REM ******* escala para la pantalla ************
250 FOR n = 1 TO u
260 LET k(n) = 30 - q + x(n)
270 LET l(n) = 10 + y(n)
280 LET r(n) = SQR ( x(n) * x(n) + y(n) * y(n) )
290 NEXT n
295 CLS
300 PLOT 210, 88 : DRAW 4,0 : PLOT 212, 86 : DRAW 0,4 : REM sol
305 PAUSE 60
310 REM ********* grafica cometa ***************.
320 FOR n = 1 TO u
330 PLOT 8 * k(n), 8 * (21 - l(n))
340 DRAW -1,0 : DRAW 0,-1 : DRAW 1,0 : DRAW 0,1
350 IF r(n) < a THEN DRAW (k(n) -212/8)*8/r(n) , (10-l(n))*8/r(n)
```

```
360 PAUSE 25
370 NEXT n
375 STOP
380 CLS
390 PLOT 210, 88 : DRAW 4,0 : PLOT 212, 86 : DRAW 0,4
400 REM ************** órbita ***************
405LET k(u+1) = k(1)
410 LET l(u+1) = l(1)
420 FOR n = 1 TO u
430 PLOT 8 * k(n), 8*(21-l(n))
440 DRAW 8*k(n+1) - 8 * k(n), 8*(21-l(n+1))-8*(21-l(n))
450 PAUSE 5
460 NEXT n
470 STOP
475 REM ********* radios vectores *************
480 FOR n = 1 TO u STEP 2
490 PLOT 8 * k(n), 8*(21-l(n))
500 DRAW 212- 8 * k(n), 88-8*(21-l(n))
510 PAUSE 25
520 NEXT n
530 REM KEPLERLEY2 ********************
```